

# NAG Library Routine Document

## G01FFF

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

### 1 Purpose

G01FFF returns the deviate associated with the given lower tail probability of the gamma distribution, via the routine name.

### 2 Specification

```
FUNCTION G01FFF (P, A, B, TOL, IFAIL)
REAL (KIND=nag_wp) G01FFF
INTEGER IFAIL
REAL (KIND=nag_wp) P, A, B, TOL
```

### 3 Description

The deviate,  $g_p$ , associated with the lower tail probability,  $p$ , of the gamma distribution with shape parameter  $\alpha$  and scale parameter  $\beta$ , is defined as the solution to

$$P(G \leq g_p : \alpha, \beta) = p = \frac{1}{\beta^\alpha \Gamma(\alpha)} \int_0^{g_p} e^{-G/\beta} G^{\alpha-1} dG, \quad 0 \leq g_p < \infty; \alpha, \beta > 0.$$

The method used is described by Best and Roberts (1975) making use of the relationship between the gamma distribution and the  $\chi^2$ -distribution.

Let  $y = 2\frac{g_p}{\beta}$ . The required  $y$  is found from the Taylor series expansion

$$y = y_0 + \sum_r \frac{C_r(y_0)}{r!} \left( \frac{E}{\phi(y_0)} \right)^r,$$

where  $y_0$  is a starting approximation

$$C_1(u) = 1,$$

$$C_{r+1}(u) = \left( r\Psi + \frac{d}{du} \right) C_r(u),$$

$$\Psi = \frac{1}{2} - \frac{\alpha - 1}{u},$$

$$E = p - \int_0^{y_0} \phi(u) du,$$

$$\phi(u) = \frac{1}{2^\alpha \Gamma(\alpha)} e^{-u/2} u^{\alpha-1}.$$

For most values of  $p$  and  $\alpha$  the starting value

$$y_{01} = 2\alpha \left( z \sqrt{\frac{1}{9\alpha} + 1} - \frac{1}{9\alpha} \right)^3$$

is used, where  $z$  is the deviate associated with a lower tail probability of  $p$  for the standard Normal distribution.

For  $p$  close to zero,

$$y_{02} = (p\alpha 2^\alpha \Gamma(\alpha))^{1/\alpha}$$

is used.

For large  $p$  values, when  $y_{01} > 4.4\alpha + 6.0$ ,

$$y_{03} = -2 \left[ \ln(1-p) - (\alpha-1) \ln\left(\frac{1}{2}y_{01}\right) + \ln(\Gamma(\alpha)) \right]$$

is found to be a better starting value than  $y_{01}$ .

For small  $\alpha$  ( $\alpha \leq 0.16$ ),  $p$  is expressed in terms of an approximation to the exponential integral and  $y_{04}$  is found by Newton–Raphson iterations.

Seven terms of the Taylor series are used to refine the starting approximation, repeating the process if necessary until the required accuracy is obtained.

## 4 References

Best D J and Roberts D E (1975) Algorithm AS 91. The percentage points of the  $\chi^2$  distribution *Appl. Statist.* **24** 385–388

## 5 Parameters

- 1: P – REAL (KIND=nag\_wp) *Input*  
*On entry:*  $p$ , the lower tail probability from the required gamma distribution.  
*Constraint:*  $0.0 \leq P < 1.0$ .
- 2: A – REAL (KIND=nag\_wp) *Input*  
*On entry:*  $\alpha$ , the shape parameter of the gamma distribution.  
*Constraint:*  $0.0 < A \leq 10^6$ .
- 3: B – REAL (KIND=nag\_wp) *Input*  
*On entry:*  $\beta$ , the scale parameter of the gamma distribution.  
*Constraint:*  $B > 0.0$ .
- 4: TOL – REAL (KIND=nag\_wp) *Input*  
*On entry:* the relative accuracy required by you in the results. The smallest recommended value is  $50 \times \delta$ , where  $\delta = \max(10^{-18}, \mathbf{machine\ precision})$ . If G01FFF is entered with TOL less than  $50 \times \delta$  or greater or equal to 1.0, then  $50 \times \delta$  is used instead.
- 5: IFAIL – INTEGER *Input/Output*  
*On entry:* IFAIL must be set to 0, –1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.  

For environments where it might be inappropriate to halt program execution when an error is detected, the value –1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, because for this routine the values of the output parameters may be useful even if IFAIL  $\neq$  0 on exit, the recommended value is –1. **When the value –1 or 1 is used it is essential to test the value of IFAIL on exit.**

*On exit:* IFAIL = 0 unless the routine detects an error or a warning has been flagged (see Section 6).

## 6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

**Note:** G01FFF may return useful information for one or more of the following detected errors or warnings.

Errors or warnings detected by the routine:

If on exit IFAIL = 1, 2, 3 or 5, then G01FFF returns 0.0.

IFAIL = 1

On entry,  $P < 0.0$ ,  
or  $P \geq 1.0$ ,

IFAIL = 2

On entry,  $A \leq 0.0$ ,  
or  $A > 10^6$ ,  
or  $B \leq 0.0$

IFAIL = 3

P is too close to 0.0 or 1.0 to enable the result to be calculated.

IFAIL = 4

The solution has failed to converge in 100 iterations. A larger value of TOL should be tried. The result may be a reasonable approximation.

IFAIL = 5

The series to calculate the gamma function has failed to converge. This is an unlikely error exit.

## 7 Accuracy

In most cases the relative accuracy of the results should be as specified by TOL. However, for very small values of  $\alpha$  or very small values of  $p$  there may be some loss of accuracy.

## 8 Further Comments

None.

## 9 Example

This example reads lower tail probabilities for several gamma distributions, and calculates and prints the corresponding deviates until the end of data is reached.

### 9.1 Program Text

```

Program g01fffe

!      G01FFF Example Program Text

!      Mark 24 Release. NAG Copyright 2012.

!      .. Use Statements ..
      Use nag_library, Only: g01fff, nag_wp
!      .. Implicit None Statement ..
      Implicit None
!      .. Parameters ..
      Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..

```

```

      Real (Kind=nag_wp)           :: a, b, p, tol, x
      Integer                      :: ifail
!      .. Executable Statements ..
      Write (nout,*) 'G01FFF Example Program Results'
      Write (nout,*)

!      Skip heading in data file
      Read (nin,*)

!      Display titles
      Write (nout,*) '      P          A          B          X'
      Write (nout,*)

!      Use default tolerance
      tol = 0.0E0_nag_wp

d_lp: Do
      Read (nin,*,Iostat=ifail) p, a, b
      If (ifail/=0) Then
         Exit d_lp
      End If

!      Calculate deviates (inverse CDF)
      ifail = -1
      x = g01fff(p,a,b,tol,ifail)
      If (ifail/=0) Then
         If (ifail/=4 .And. ifail/=5) Then
            Exit d_lp
         End If
      End If

!      Display results
      Write (nout,99999) p, a, b, x
End Do d_lp

99999 Format (1X,3F8.3,F10.3,A,I1)
End Program g01fffe

```

## 9.2 Program Data

```

G01FFF Example Program Data
0.0100  1.0 20.0           :P A B
0.4279  7.5 0.1           :P A B
0.8694  45.0 10.0         :P A B

```

## 9.3 Program Results

```

G01FFF Example Program Results

      P          A          B          X
0.010  1.000  20.000    0.201
0.428  7.500   0.100    0.670
0.869  45.000  10.000   525.979

```

---